

UR1987-51

1987/51. SLIPCIRC - a GW-BASIC program for Bishop's simplified slip circle stability analysis on an IBM-compatible micro-computer.

B. D. Weldon

Abstract

A slope stability analysis program based on Bishop's simplified method has been written in GW-BASIC for an IBM-compatible micro-computer. The use of the program is described and a listing with detailed comments is provided.

INTRODUCTION

Moon (1984) introduced readers to slope stability analysis using a Hewlett-Packard 41C programmable calculator. His program has been modified and written in GW-BASIC so as to run on an IBM-compatible micro-computer. This report comments on the program and enhancements now provided.

The major restraint of the program is that the subsurface cannot be stratified. The entire earth mass under consideration is assigned one set of parameters (i.e. density, cohesion and friction) for each particular analysis. Pore pressures are considered as the pore pressure ratio ($ru = \gamma_w \cdot h / \gamma_d$). There is no limit, apart from available computer memory, on the number of slices permitted. In reality, only rarely will more than 20 slices be required. The slices are of uniform width for each particular analysis.

A full listing of the program, including detailed comments, is provided as Appendix 1. The program may be activated on the Engineering Geology and Hydrology micro-computer by selecting MODELS and SLIP CIRCLE from the start-up menu displays. Alternatively, the GW-BASIC environment may be invoked and the program SLIPCIRC.BAS loaded from the directory MODELS on the hard disk.

DATA INPUT

The initial screen display requests a complete set of parameters necessary to define a slip circle for analysis (fig. 1). Data are entered by typing the relevant values to a screen prompt. The back-space, insert and delete functions may be used to correct mistakes made as the data are being entered. To complete a data entry, the return key must be depressed. Once this key is depressed, the particular entry cannot be altered prior to calculation. It is necessary to complete the data input, allow the calculation to be made, and the result displayed. The OPTIONS menu will then be displayed and alterations can be made to any of the input data via the change data option.

CALCULATIONS

When the last data value has been provided, the program calculates and displays the new factor of safety. The expression evaluated is:

$$F = \frac{\sum ((c' \cdot b + (W - ub) \tan \phi') (1/m\alpha))}{\sum W \cdot \sin \alpha}$$

where $m\alpha = \cos (1 + (\tan\alpha \cdot \tan\phi') / F)$
 c' = cohesion
 b = width of slice
 W = weight of a slice
 u = pore pressure ratio
 ϕ' = friction angle
 α = slope of failure plane
 F = Factor of Safety

As F appears on both sides of the expression, it is most convenient to solve for F in an iterative manner.

OPTIONS MENU

Immediately the new Factor of Safety has been calculated, the OPTIONS menu will be displayed (fig. 2). An option is selected by typing the first letter of its name. The options are:

Iterate: Entering I or i at the option menu prompt replaces the original factor of safety with the just calculated value. A status message is displayed and the calculation routine is re-entered. The new Factor of Safety will be displayed and program control returns to the options menu.

Change data: A new menu (fig. 3) replaces the options menu when C or c is entered at the option menu prompt. Parameters may be changed by simply typing the first letter of the required parameters name. The existing value is erased and a new value is requested. To complete data entry, the return key must be depressed. As many parameters as desired may be changed before typing O (for Ok to re-calculate). This causes the program to re-enter the calculation routine, display the new Factor of Safety, and returns control to the option menu.

When changing the number of slices, it is assumed that the end points will not change but that a new ground shape (i.e. Y co-ordinate for the midpoint of each slice) will be needed along with a new Factor of Safety.

To allow for the modelling of tension cracks, the end points may be changed (new values for each of the two co-ordinate pairs will be requested).

It is necessary to enter O or o to exit from the change menu.

End: Selecting E or e at the option menu prompt closes any files that may have been opened and returns control to the MS-DOS environment.

Print data: Entering a P or p at the option menu prompt will cause the parameters used in the calculation to be sent to the line printer (provided it is switched on). A status message is displayed during this process. Initially, the ground shape (i.e. the Y co-ordinates for the mid-points of each slice) values are printed followed by each of the other parameters in tabular format. On subsequent print requests, the ground shape data are only printed should they have been changed via the options menu.

A typical print-out is given in Appendix 2.

Save data: Entering a S or s at the options menu prompt will cause all the parameters used in the calculation to be written to disk. The first time this option is selected, a new screen display seeking information about the file specification (fig. 4) is presented. Default values for the file specification are provided on this screen. The return key need only be depressed to accept each of the default values.

New values may be typed over the defaults. To complete entering a new value, the return key must be depressed. The file specification will be displayed briefly before returning to the original screen. A status message is displayed as the data is written to the disk. The data is written in comma separated ASCII format (i.e. a comma is placed between consecutive values). Program control is returned to the options menu.

Range: Typing a R or r at the options menu prompt will allow the user to specify a range of density, pore pressure, cohesion, Y circle, or friction for which calculations will be made, the results printed, and all parameters written to disk automatically. When this option is selected the options menu is replaced by the RANGE FOR menu (fig. 5).

A parameter for range calculations is selected by typing the first letter of its name. This causes the range for menu to be replaced by the range request menu (fig. 6). The start, end and step values are entered, depressing the return key after each to accept the value. The step value is the amount by which the parameter will be incremented between consecutive calculations.

After entering the step value, a status message will be displayed. The start value of the chosen parameter will be displayed and the calculation routine entered until the old and new Factor of Safety converge on the second decimal place. The parameters are then printed to the line printer and written to disk. File specification is requested if not previously given. The new Factor of Safety is displayed to the screen, along with the incremented value of the chosen parameter. When calculations have been completed for the specified range, the program control returns to the options menu.

REFERENCE

MOON, A. T. 1984. Bishop's simplified slope stability analysis with a Hewlett-Packard 41C calculator. *Unpubl. Rep. Dep. Mines Tasm.* 1984/51.

[27 October 1987]

STABILITY ANALYSIS: SLIP CIRCLE

END POINTS:	X lhs = 4	GROUND SHAPE: Slice	Y co-ord
	Y lhs = 4	1	5.5
	X rhs = 24	2	7.5
	Y rhs = 15	3	9
		4	11.5
		5	14.5

Number of Slices = 5
 Density (kN/cub m) = 20
 Cohesion (kPa) = 10
 Friction (degrees) = 30
 Ru = .5
 Y Circle (m) = 39
 Safety Factor = 1

Figure 1.

STABILITY ANALYSIS: SLIP CIRCLE

END POINTS:	X lhs = 4	GROUND SHAPE: Slice	Y co-ord
	Y lhs = 4	1	5.5
	X rhs = 24	2	7.5
	Y rhs = 15	3	9
		4	11.5
		5	14.5

Number of Slices = 5
 Density (kN/cub m) = 20
 Cohesion (kPa) = 10
 Friction (degrees) = 30
 Ru = .5
 Y Circle (m) = 39
 Safety Factor = 1

New Factor of Safety = 1.15

OPTIONS:	Iterate	Print data
	Change data	Save data
	End	Range

(Enter I,C,P,S,E or R)

Figure 2.

STABILITY ANALYSIS: SLIP CIRCLE

END POINTS:	X lhs = 4	GROUND SHAPE: Slice	Y co-ord
	Y lhs = 4	1	5.5
	X rhs = 24	2	7.5
	Y rhs = 15	3	9
		4	11.5
		5	14.5

Number of Slices = 5
 Density (kN/cub m) = 20
 Cohesion (kPa) = 10
 Friction (degrees) = 30
 Ru = .5
 Y Circle (m) = 39
 Safety Factor = 1.15

CHANGE: Density End Points
 Cohesion Number of Slices
 Friction Ground Shape
 Ru Safety Factor
 Y Circle Ok to re-calculate

(Enter D,C,F,R,Y,E,N,G,S or O)

Figure 3.

PROVIDE DISK DRIVE A:
 path-name \
 and file-name STABIL.DAT

File specification is
 A:\STABIL.DAT

Figure 4.

STABILITY ANALYSIS: SLIP CIRCLE

END POINTS:	X lhs = 4	GROUND SHAPE: Slice	Y co-ord
	Y lhs = 4	1	5.5
	X rhs = 24	2	7.5
	Y rhs = 15	3	9
		4	11.5
		5	14.5

Number of Slices = 5
 Density (kN/cub m) = 20
 Cohesion (kPa) = 10
 Friction (degrees) = 30
 Ru = .5
 Y Circle (m) = 39
 Safety Factor = 1.15

RANGE FOR: Density Ru
 Cohesion Y Circle
 Friction

(Enter D,C,F,R OR Y)

Figure 5.

STABILITY ANALYSIS: SLIP CIRCLE

END POINTS:	X lhs = 4	GROUND SHAPE: Slice	Y co-ord
	Y lhs = 4	1	5.5
	X rhs = 24	2	7.5
	Y rhs = 15	3	9
		4	11.5
		5	14.5

Number of Slices = 5
 Density (kN/cub m) = 20
 Cohesion (kPa) = 10
 Friction (degrees) = 30
 Ru = .5
 Y Circle (m) = 39
 Safety Factor = 1.15

Start value of Ru = 0
 End value of Ru = .5
 Calculate in steps of = .1

Figure 6.

APPENDIX 1

Program SLIPCIRC

```

10 REM SLIPCIRC: a slip circle stability analysis based on BISHOP'S method
20 '
30 '          version #3.0      APRIL 87      B D WELDON
40 '
50 '
60 '
70 SCREEN 0,0,0,0          : ' set screen 0 as active
80 KEY OFF                 : ' turn function key display off
90 CLS                     : ' clear screen display
100 COLOR 1,0              : ' set back/fore-ground colors
110 LOCATE 1,20            : ' find line 1, col 20 on screen
120 PRINT "STABILITY ANALYSIS: SLIP CIRCLE" : ' display title to screen
130 COLOR 7,0:LOCATE 3,1:PRINT "END POINTS:" : ' display to screen
140 J =1:K=11              : ' set counter start/stop values
150 GOSUB 1020             : ' DATA INPUT
160 DIM Y(NS),TH(NS),U(NS),W(NS) : ' dimension up arrays
170 PT=0:AC=0:TX=0        : ' set flag values
180 LOCATE 3,39            : ' set screen position
190 PRINT "GROUND SHAPE: Slice      Y co-ord" : ' display to screen
200 VP=4:HP=55:J=1:K=NS   : ' set screen/counter parameters
210 GOSUB 1290             : ' get GROUND SHAPE values
220 GOSUB 1400             : ' calculate SAFETY FACTOR
230 '
240 ' *****
250 ' DISPLAY CALCULATED FACTOR OF SAFETY
260 ' *****
270 '
280 LOCATE 16,4:COLOR 7,0 : ' set screen display parameters
290 PRINT "New Factor of Safety = "; : ' display to screen
300 PRINT USING "#.##";FS  : ' display formatted result
310 '
320 ' *****
330 ' CHECK FLAGS
340 ' *****
350 '
360 IF AC=0 THEN GOSUB 2020 : ' check flags, display OPTIONS
370 IF AC=1 AND ACC=1 THEN 6380 : ' check flags, data PRINTED
380 IF AC=1 AND ACC=0 THEN 6400 : ' check flags, data SAVED
390 '
400 ' *****
410 ' CHECK KEYBOARD RESPONSE
420 ' *****
430 '
440 GOSUB 2160             : ' check keyboard for response
450 AN%=INSTR("IiCcPpSsEeRr",AN%) : ' valid response range
460 IF AN%<1 OR AN%>12 THEN 410 : ' check validity of response
470 ON AN% GOTO 2240,2240,2360,2360,3020,3020 : ' goto selected response
      ,4020,4020,5020,5020,6020,6020
480 '
490 ' *****
500 ' DATA BANK
510 ' *****
520 '
530 DATA "X lhs = ","###.#",3,19 : ' DATA BANK
540 DATA "Y lhs = ","###.#",4,19
550 DATA "X rhs = ","###.#",5,19
560 DATA "Y rhs = ","###.#",6,19
570 DATA "Number of Slices = ","##",8,8
580 DATA "Density (kN/cub m) = ","##.#",9,6
590 DATA "Cohesion (kPa) = ","##.#",10,10
600 DATA "Friction (degrees) = ","##.#",11,6
610 DATA "Ru = ","#.##",12,22
620 DATA "Y Circle (m) = ","###.#",13,12
630 DATA "Safety Factor = ","#.##",14,11
1000

```

```

1010 ' *****
1020 ' DATA INPUT
1030 ' *****
1040 '
1050 FOR I=J TO K : ' start counting parameters
1060 READ VN#,PU#,VP,HP : ' read data
1070 LOCATE VP,HP:PRINT VN#,:INPUT "",VV : ' request value
1080 ON I GOSUB 1160,1170,1180,1190,1200,
1210,1220,1230,1240,1250,1260 : ' assign input to parameter
1090 NEXT I : ' next parameter
1100 RETURN : ' finished here
1110 '
1120 ' *****
1130 ' ASSIGN PARAMETER TO INPUT
1140 ' *****
1150 '
1160 XL=VV:RETURN : ' X lhs
1170 YL=VV:RETURN : ' Y lhs
1180 XR=VV:RETURN : ' X rhs
1190 YR=VV:RETURN : ' Y rhs
1200 NS=VV:RETURN : ' Number of Slices
1210 GA=VV:RETURN : ' Density
1220 CO=VV:RETURN : ' Cohesion
1230 PH=VV:RETURN : ' Angle of Internal Friction
1240 RU=VV:RETURN : ' Pore Pressure Ratio
1250 YC=VV:RETURN : ' Y Circle
1260 F1=VV:RETURN : ' Factor of Safety
1270 '
1280 ' *****
1290 ' GROUND SHAPE DATA INPUT
1300 ' *****
1310 '
1320 FOR I=J TO K : ' start counting slices
1330 LOCATE VP,HP:PRINT I : ' display slice number
1340 LOCATE VP,HP+10:INPUT "",Y(I) : ' request Y co-ordinate
1350 VP=VP+1 : ' increment screen line counter
1355 IF I=20 THEN VP=4
1360 NEXT I : ' next slice
1370 RETURN : ' finished here
1380 '
1390 ' *****
1400 ' CALCULATIONS
1410 ' *****
1420 '
1430 XA=(XL+XR)/2:YA=(YL+YR)/2:XD=XL-XR:YD=YL-YR: ' calculate geometry
1440 XC=XA-(YC-YA)*YD/XD : '
1450 CI=SQR((XC-XL)*(XC-XL)+(YC-YL)*(YC-YL)) : '
1460 DF=0:RF=0:U=0:W=0:J=1:K=NS : ' initialise accumulators
1470 FOR I=J TO K : ' start counting slices
1480 T1=(XR-XL)/NS:T2=I-.5:T3=XL-XC:
T4=360/(2*3.141593):T5=2*3.141593/360 : ' calculate forces
1490 TH=(T1*T2+T3)/CI:TEMP=TH/SQR(1-TH*TH) : '
1500 TH(I)=ATN(TEMP)*T4 : '
1510 W(I)=((Y(I)+(CI*CO*(TH(I)*T5)))-YC)*GA : '
1520 W=W+SIN(TH(I)*T5)*W(I) : '
1530 DF=DF+SIN(TH(I)*T5)*W(I) : '
1540 U(I)=(1-RU)*W(I)*TAN(PH*T5)+CO : '
1550 V=(SIN(TH(I)*T5)*TAN(PH*T5)/F1)+
COS(TH(I)*T5)
1560 RF=RF+(U(I)/V) : '
1570 NEXT I : ' next slice
1580 FS=RF/DF : ' calculate Factor of Safety
1590 RETURN : ' finished here
2000 '
2010 ' *****
2020 ' OPTION MENU

```

```

2030 ' *****
2040 '
2050 LOCATE 18,5:COLOR 1,0:PRINT "OPTIONS:" : ' display OPTIONS menu
2060 LOCATE 18,16:COLOR 4,6:PRINT "I":
LOCATE 18,17:COLOR 7,0:PRINT "terate" : ' iterate
2070 LOCATE 18,35:COLOR 4,6:PRINT "P":
LOCATE 18,36:COLOR 7,0:PRINT "rint data" : ' print data
2080 LOCATE 19,16:COLOR 4,6:PRINT "C":
LOCATE 19,17:COLOR 7,0:PRINT "hange data" : ' change data
2090 LOCATE 19,35:COLOR 4,6:PRINT "S":
LOCATE 19,36:COLOR 7,0:PRINT "ave data" : ' save data to disk
2100 LOCATE 20,16:COLOR 4,6:PRINT "E":
LOCATE 20,17:COLOR 7,0:PRINT "nd" : ' end session
2110 LOCATE 20,35:COLOR 4,6:PRINT "R":
LOCATE 20,36:COLOR 7,0:PRINT "ange" : ' range of data for auto-calc
2120 LOCATE 22,18:COLOR 3,0:
PRINT "(Enter I,C,P,S,E or R)" : ' display valid responses
2130 RETURN : ' finished here
2140 '
2150 ' *****
2160 ' KEYBOARD RESPONSE
2170 ' *****
2180 '
2190 AN$=INKEY$ : ' assign keyboard response
2200 IF AN$="" THEN 2190 : ' check keyboard for response
2210 RETURN : ' finished here
2220 '
2230 ' *****
2240 ' ITERATE OPTION
2250 ' *****
2260 '
2270 LOCATE 22,16:COLOR 7,0:PRINT SPC(34) : ' clear prompt line from screen
2280 LOCATE 22,18:COLOR 7,0:
PRINT "RE-CALCULATING" : ' display status message
2290 F1=FS : ' re-assign Factor of Safety
2300 LOCATE 14,26:COLOR 7,0:PRINT SPC(24) : ' remove old value from screen
2310 LOCATE 14,27:PRINT USING "#.##";FS : ' display new value to screen
2320 GOSUB 1400 : ' now re-calculate FS
2330 GOTO 250 : ' and display to screen
2340 '
2350 ' *****
2360 ' CHANGE PARAMETER(S)
2370 ' *****
2380 '
2390 HD=1 : ' set flag for header print
2400 FOR I=16 TO 23:LOCATE I,1:PRINT SPC(50):
NEXT I : ' erase part of screen display
2410 LOCATE 18,3:COLOR 1,0:PRINT "CHANGE:" : ' display CHANGE menu
2420 LOCATE 18,13:COLOR 4,6:PRINT "D":
LOCATE 18,14:COLOR 7,0:PRINT "ensity" : ' density
2430 LOCATE 18,33:COLOR 4,6:PRINT "E":
LOCATE 18,34:COLOR 7,0:PRINT "nd Points" : ' end points
2440 LOCATE 19,13:COLOR 4,6:PRINT "C":
LOCATE 19,14:COLOR 7,0:PRINT "hesion" : ' cohesion
2450 LOCATE 19,33:COLOR 4,6:PRINT "N":LOCATE
19,34:COLOR 7,0:PRINT "umber of Slices" : ' number of slices
2460 LOCATE 20,13:COLOR 4,6:PRINT "F":
LOCATE 20,14:COLOR 7,0:PRINT "riction" : ' friction
2470 LOCATE 20,33:COLOR 4,6:PRINT "G":
LOCATE 20,34:COLOR 7,0:PRINT "round Shape" : ' ground shape
2480 LOCATE 21,13:COLOR 4,6:PRINT "R":
LOCATE 21,14:COLOR 7,0:PRINT "u" : ' Ru
2490 LOCATE 21,33:COLOR 4,6:PRINT "S":
LOCATE 21,34:COLOR 7,0:PRINT "afety Factor": ' Safety Factor
2500 LOCATE 22,13:COLOR 4,6:PRINT "Y":
LOCATE 22,14:COLOR 7,0:PRINT " Circle" : ' Y circle

```

```

2510 LOCATE 22,33:COLOR 4,6:PRINT "0":
      LOCATE 22,34:COLOR 7,0:
      PRINT "k to re-calculate"           : ' re-calculate now
2520 LOCATE 24,13:COLOR 3,0:
      PRINT "(Enter D,C,F,R,Y,E,N,G,S or 0)": : ' display valid responses
2530 GOSUB 2160                           : ' get keyboard response
2540 AN% = INSTR("DdCcFfRrYyNnEeGgSsOo",AN%) : ' valid responses
2550 IF AN%<1 OR AN%>20 THEN 2530         : ' check validity of response
2560 LOCATE 24,13:COLOR 7,0:PRINT SPC(37) : ' clear prompt line
2570 ON AN% GOTO 2580,2580,2590,2590,2600,2600,
      2610,2610,2620,2620,2630,2630,2670,2670,
      2740,2740,2860,2860,2870,2870       : ' goto selected response
2580 VP=9:HP=27:GOSUB 2940:GOSUB 1210:GOTO 2520 : ' change density
2590 VP=10:HP=27:GOSUB 2940:GOSUB 1220:GOTO 2520 : ' change cohesion
2600 VP=11:HP=27:GOSUB 2940:GOSUB 1230:GOTO 2520 : ' change friction
2610 VP=12:HP=27:GOSUB 2940:GOSUB 1240:GOTO 2520 : ' change Ru
2620 VP=13:HP=27:GOSUB 2940:GOSUB 1250:GOTO 2520 : ' change Y circle
2630 VP=8:HP=27:L=NS:GOSUB 2940:GOSUB 1200      : ' change number of slices
2640 ERASE Y,TH,U,W                             : ' erase old arrays
2650 DIM Y(NS),TH(NS),U(NS),W(NS)               : ' dimension up new arrays
2660 GOTO 2740                                   : ' now get new ground shape
2670 VP=3:HP=27:J=i:K=4                         : ' set parameters
2680 FOR I=J TO K                               : ' start counting end points
2690     GOSUB 2940                             : ' re-fresh screen display
2700     ON I GOSUB 1160,1170,1180,1190         : ' assign new end point values
2710     VP=VP+1                                : ' increment screen line counter
2720 NEXT I                                     : ' next end point
2730 GOTO 2520                                  : ' finished! display prompt line
2740 VP=4:HP=65:J=1:K=NS:PT=0                 : ' set parameters
2750 FOR I=J TO L                               : ' start counting old slices
2760     LOCATE VP,HP-12:PRINT SPC(24)         : ' erase old ground shape value
2770     VP=VP+1                                : ' increment screen line counter
2775     IF I=21 THEN I=L                      : '
2780 NEXT I                                     : ' next old slice
2790 VP=4                                       : ' re-set screen line display
2800 FOR I=J TO K                               : ' start counting new slices
2810     LOCATE VP,HP-10:PRINT I               : ' display slice number
2820     GOSUB 2940                             : ' remove old value and get new
2830     Y(I)=VV                                : ' assign new value for slice
2840     VP=VP+1                                : ' increment screen line counter
2845     IF I=20 THEN VP=4                    : '
2850 NEXT I                                     : ' nex new slice
2860 VP=14:HP=27:GOSUB 2940:GOSUB 1260:
      GOTO 2520                                 : ' change safety factor
2870 FOR I=18 TO 24                             : ' start counting screen lines
2880     LOCATE I,1                             : ' goto screen position
2890     PRINT SPC(50)                          : ' erase portion of line
2900 NEXT I                                     : ' next screen line
2910 GOTO 220                                   : ' calculate safety factor
2920 '
2930 ' *****
2940 ' REFRESH SCREEN DISPLAY
2950 ' *****
2960 '
2970 LOCATE VP,HP:PRINT SPC(10)                : ' erase existing value
2980 LOCATE VP,HP:INPUT "",VV                 : ' request a new value
2990 RETURN                                    : ' finished here
3000 '
3010 ' *****
3020 ' PRINT OPTION
3030 ' *****
3040 '
3050 LOCATE 22,1:PRINT SPC(50):LOCATE 22,18    : ' delete prompt line
3060 COLOR 3,0:PRINT "PRINTING DATA"         : ' display status message
3070 PT=PT+1:IF PT>1 THEN 3080 ELSE 3090      : ' check printer flag
3080 IF HD=1 THEN 3170 ELSE 3200              : ' check if header required

```

```

3090 LPRINT :LPRINT                               : ' advance printer two lines
3100 LPRINT "GROUND SHAPE: Slice      Y co-ord"    : ' print header information
3110 J=1:K=NS                                       : ' set parameters
3120 FOR I=J TO K                                   : ' start counting slices
3130     LPRINT SPC(16);:LPRINT USING "##";I;      : ' print slice number
3140     LPRINT SPC(7);                             : ' print seven spaces
3150     LPRINT USING "###.#";Y(I)                 : ' print Y co-ordinate
3160 NEXT I                                         : ' next slice
3170 LPRINT                                         : ' advance printer 1 line
3180 LPRINT " Xlhs Ylhs Xrhs Yrhs NS Gamm";       : ' print header information
3190 LPRINT " Cohs Phi Ru Yc F1 FS"              : '
3200 RESTORE:J=1:K=11                              : ' set parameters
3210 FOR I=J TO K                                   : ' start counting parameters
3220     READ VN$,VU$,VP,HP                         : ' read data
3230     ON I GOSUB 3330,3340,3350,3360,3370,      : ' goto appropriate parameter
           3380,3390,3400,3410,3420,3430
3240     LPRINT USING VU$;VV$;:LPRINT SPC(1);      : ' print parameter
3250 NEXT I                                         : ' next parameter
3260 LPRINT USING "#.###";FS                       : ' print Factor of Safety
3270 HD=0:GOTO 360                                 : ' return to main menu
3280 '
3290 ' *****
3300 ' PROVIDE PARAMETER VALUE
3310 ' *****
3320 '
3330 VV=XL:RETURN                                  : '
3340 VV=YL:RETURN                                  : '
3350 VV=XR:RETURN                                  : '
3360 VV=YR:RETURN                                  : '
3370 VV=NS:RETURN                                  : '
3380 VV=GA:RETURN                                  : '
3390 VV=CO:RETURN                                  : '
3400 VV=PH:RETURN                                  : '
3410 VV=RU:RETURN                                  : '
3420 VV=YC:RETURN                                  : '
3430 VV=F1:RETURN                                  : '
4000 '
4010 ' *****
4020 ' SAVE DATA OPTION
4030 ' *****
4040 '
4050 TX=TX+1: IF TX>1 THEN 4060 ELSE GOSUB 4250  : ' check for file specification
4060 LOCATE 22,1:COLOR 7,0:PRINT SPC(50)          : ' delete prompt line
4070 LOCATE 22,16:COLOR 3,0:PRINT "SAVING DATA"  : ' display status message
4080 OPEN FILESPEC$ FOR APPEND AS #1              : ' open file, append mode
4090 PRINT #1,USING "###.#,";XL,YL,XR,YR,YC;      : ' write to disk XL,YL,XR,YR,YC
4100 PRINT #1,USING "##";NS;:PRINT #1,",";        : ' write to disk NS
4110 PRINT #1,USING "##.#,";GA,CO,PH;            : ' write to disk GA,CO,PH
4120 PRINT #1,USING "#.##,";RU,F1,FS;            : ' write to disk RU,F1,FS
4130 FOR I=1 TO NS                                 : ' start counting slices
4140     PRINT #1,USING "###.#,";Y(I);            : ' write to disk Y co-ord
4150 NEXT I                                         : ' increment counter
4160 PRINT #1, CHR$(13)                            : ' write carriage return to disk
4170 CLOSE #1                                       : ' close file
4180 LOCATE 22,1:PRINT SPC(50)                    : ' delete status message
4190 GOTO 360                                       : ' return to OPTIONS MENU
4200 '
4210 ' *****
4220 ' FILE SPECIFICATION
4230 ' *****
4240 '
4250 DDRIVE$="A:"                                    : ' set default drive
4260 DPATH$="\ "                                    : ' set default path
4270 DFILE$="STABIL"                                : ' set default file-name
4280 DEXTENSION$=".DAT"                             : ' set default extension
4290 SCREEN 0,0,1,1                                : ' select screen 1 as active

```

```

4300 CLS:LN=12:CN=20:FG=7:BG=0:FC=3:BC=0      : ' set screen parameters
4310 LOCATE LN,CN:COLOR FG,BG:                 : ' display to screen
      PRINT "PROVIDE DISK DRIVE "
4320 LOCATE LN,CN+20:COLOR FC,BC:             : ' display default disk drive
      PRINT DDRIVE$
4330 LOCATE LN,CN+20:COLOR FG,BG:             : ' request disk drive
      INPUT "",DRIVESPEC$
4340 LOCATE LN+1,CN+9:COLOR FG,BG:            : ' display to screen
      PRINT "path-name"
4350 LOCATE LN+1,CN+20:COLOR FC,BC:           : ' display default path
      PRINT DPATH$
4360 LOCATE LN+1,CN+20:COLOR FG,BG:           : ' request path-name
      INPUT "",PATHSPEC$
4370 LOCATE LN+2,CN+5:COLOR FG,BG:            : ' display to screen
      PRINT "and file-name"
4380 LOCATE LN+2,CN+20:COLOR FC,BC:           : ' display default filename
      PRINT DFILE$+DEXTENSION$
4390 LOCATE LN+2,CN+20:COLOR FG,BG:           : ' request file-name
      INPUT "",FILENAME$
4400 IF LEN(DRIVESPEC$)=0                      : ' default drive required
      THEN DRIVESPEC$=DDRIVE$
4410 IF LEN(DRIVESPEC$)=1 AND DRIVESPEC$<>":": : ' colon(:) required
      THEN DRIVESPEC$=DRIVESPEC$+": "
4420 IF LEN(PATHSPEC$)=0                      : ' default path required
      THEN PATHSPEC$=DPATH$
4430 IF RIGHT$(PATHSPEC$,1)<>"\"              : ' backslash required
      THEN PATHSPEC$=PATHSPEC$+"\ "
4440 IF LEFT$(PATHSPEC$,1)<>"\"              : ' backslash required
      THEN PATHSPEC$="\ "+PATHSPEC$
4450 L%=LEN(FILENAME$):                       : ' determine length of filename
      IF L%=0 THEN 4460 ELSE 4470
4460 FILENAME$=DFILE$:GOTO 4550                : ' default filename required
4470 FOR I=1 TO L%                             : ' start counting characters
4480     IF RIGHT$(LEFT$(FILENAME$,I),1)=".":   : ' check for extension delimiter
      THEN NEWFILE$=LEFT$(FILENAME$,I-1)
      ELSE 4530
4490     IF LEN(NEWFILE$)>8                    : ' limit filename to 8 chars
      THEN NEWFILE$=LEFT$(NEWFILE$,8)
4500     NEWEXTN$=RIGHT$(FILENAME$,L%-I)       : ' rest is extension
4510     IF LEN(NEWEXTN$)>3                    : ' limit extension to 3 chars
      THEN NEWEXTN$=LEFT$(NEWEXTN$,3)
4520     FILENAME$=NEWFILE$+"."+NEWEXTN$:      : ' assemble filename
      GOTO 4560
4530 NEXT I                                    : ' next character
4540 IF LEN(FILENAME$)>8                      : ' limit filename to 8 chars
      THEN FILENAME$=LEFT$(FILENAME$,8)
4550 FILENAME$=FILENAME$+DEXTENSION$           : ' assemble file name
4560 FILESPEC$=DRIVESPEC$+PATHSPEC$+FILENAME$ : ' assemble file specification
4570 LOCATE LN+7,CN:COLOR FG,BG:              : ' display to screen
      PRINT "      File specification is"
4580 LOCATE LN+8,CN+10:COLOR FC,BC:           : ' display file specification
      PRINT FILESPEC$
4590 FOR J=1 TO 1500:NEXT J                   : ' time delay for display
4600 SCREEN 0,0,0,0                           : ' return to screen 0
4610 RETURN                                    : ' finished here
5000 '
5010 ' *****
5020 ' END OPTION
5030 ' *****
5040 '
5050 COLOR 7,0:CLOSE:SYSTEM                   : ' turn function keys on, clear
                                          : screen and end program

6000 '
6010 ' *****
6020 ' AUTO-RANGE CALCULATION OPTION
6030 ' *****

```

```

6040
6050 HD=1 : ' set flag for header print
6060 FOR I=16 TO 22 : ' start counting line numbers
6070 LOCATE I,1:COLOR 7,0:PRINT SPC(50) : ' delete line from screen
6080 NEXT I:AC=1 : ' next line number, set flag
6090 LOCATE 18,3:COLOR 1,0:PRINT "RANGE FOR:" : ' request parameter name
6100 LOCATE 18,16:COLOR 4,6:PRINT "D":
LOCATE 18,17:COLOR 7,0:PRINT "ensity" : ' display Density
6110 LOCATE 18,35:COLOR 4,6:PRINT "R":
LOCATE 18,36:COLOR 7,0:PRINT "u" : ' display Ru
6120 LOCATE 19,16:COLOR 4,6:PRINT "C":
LOCATE 19,17:COLOR 7,0:PRINT "ohesion" : ' display Cohesion
6130 LOCATE 19,35:COLOR 4,6:PRINT "Y":
LOCATE 19,36:COLOR 7,0:PRINT " Circle" : ' display Y circle
6140 LOCATE 20,16:COLOR 4,6:PRINT "F":
LOCATE 20,17:COLOR 7,0:PRINT "riction" : ' display Friction
6150 LOCATE 22,18:COLOR 3,0:
PRINT "(Enter D,C,F,R OR Y)" : ' display valid responses
6160 GOSUB 2160 : ' check keyboard for response
6170 AN%=INSTR("DdCcFfRrYy",AN%) : ' check validity of response
6180 IF AN%<1 OR AN%>10 THEN 6160 : ' if invalid go again
6190 FOR I=16 TO 22 : ' start counting screen lines
6200 LOCATE I,1:COLOR 7,0:PRINT SPC(50): : ' erase portion of line
6210 NEXT I : ' next screen line
6220 ON AN% GOSUB 6470,6470,6480,6480,
6490,6490,6500,6500,6510,6510 : ' goto desired response
6230 READ VN$,PU$,OVP,HP : ' read data
6240 LOCATE 18,5:COLOR 7,0:
PRINT "Start value of "VN$;:
INPUT "",SV : ' request range starting value
6250 LOCATE 19,5:COLOR 7,0:
PRINT " End value of "VN$;:
INPUT "",EV : ' request range end value
6260 LOCATE 20,15:COLOR 7,0:
INPUT " Calculate in steps of = ",INCR : ' request increment value
6270 LOCATE 22,10:COLOR 3,0:
PRINT "AUTO-RANGE CALCULATIONS IN PROGRESS":
AC=1 : ' display status message,
: ' start counting range increm.
6280 FOR IV=SV TO EV STEP INCR:VV=IV
6290 ON AN% GOSUB 1210,1210,1220,1220,
1230,1230,1240,1240,1250,1250 : ' get appropriate data
6300 LOCATE OVP,27:PRINT SPC(10) : ' erase old value
6310 LOCATE OVP,27:PRINT USING PU$;VV : ' display current value
6320 GOSUB 1400 : ' now for calculations
6330 IF INT(F1*100)<>INT(FS*100)
THEN 6340 ELSE 6370 : ' check for convergence
6340 LOCATE 22,1:COLOR 7,0:PRINT SPC(50) : ' clear prompt line
6350 LOCATE 22,10:COLOR 3,0:PRINT
"AUTO-RANGE CALCULATIONS IN PROGRESS" : ' display status message
6360 F1=FS:GOTO 6320 : ' not yet converged, go again
6370 ACC=1:GOTO 3020 : ' converged, print results
6380 ACC=0:GOTO 4020 : ' save data to disk
6390 LOCATE 14,27:COLOR FC,BC:
PRINT SPC(10) : ' erase old safety factor
6400 LOCATE 14,27:COLOR FC,BC:
PRINT USING "#.##";FS : ' display new safety factor
6410 F1=1:FS=1 : ' re-set FS to unity
6420 NEXT IV : ' next range increment
6430 FOR I=16 TO 22:LOCATE I,1:COLOR 7,0
6440 PRINT SPC(50) : ' start counting screen lines
: ' erase screen line
6450 NEXT I : ' next screen line
6460 AC=0:GOTO 330 : ' return to OPTIONS MENU
6470 RESTORE 580:RETURN : ' read data
6480 RESTORE 590:RETURN
6490 RESTORE 600:RETURN
6500 RESTORE 610:RETURN
6510 RESTORE 620:RETURN

```

APPENDIX 2

GROUND SHAPE: Slice	Y co-ord
1	5.5
2	7.5
3	9.0
4	11.5
5	14.5

Xlhs	Ylhs	Xrhs	Yrhs	NS	Gamm	Cohs	Phi	Ru	Yc	F1	FS
4.0	4.0	24.0	15.0	5	20.0	10.0	30.0	0.00	39.0	1.90	1.904
4.0	4.0	24.0	15.0	5	20.0	10.0	30.0	0.10	39.0	1.76	1.762
4.0	4.0	24.0	15.0	5	20.0	10.0	30.0	0.20	39.0	1.61	1.618
4.0	4.0	24.0	15.0	5	20.0	10.0	30.0	0.30	39.0	1.47	1.475
4.0	4.0	24.0	15.0	5	20.0	10.0	30.0	0.40	39.0	1.33	1.333
4.0	4.0	24.0	15.0	5	20.0	10.0	30.0	0.50	39.0	1.18	1.188

Xlhs	Ylhs	Xrhs	Yrhs	NS	Gamm	Cohs	Phi	Ru	Yc	F1	FS
4.0	4.0	24.0	15.0	5	20.0	10.0	30.0	0.50	24.0	1.00	0.969
4.0	4.0	24.0	15.0	5	20.0	10.0	30.0	0.50	24.0	0.97	0.962
4.0	4.0	24.0	15.0	5	20.0	10.0	30.0	0.50	24.0	0.96	0.961